query: #Persons with more Wikiquote articles than Wikipedia articles. 
  #I have limited the search to persons with occupation (P106) author (Q482980) to make the query possible without timeout.
  SELECT ?item ?itemLabel ?wikipedia_articles ?wikiquote_articles
  WITH
  {
    SELECT ?item (COUNT(DISTINCT ?wikipedia_article) AS ?wikipedia_articles)
          (COUNT(DISTINCT ?wikiquote_article) AS ?wikiquote_articles)
    WHERE
    {
      ?item wdt:P31 wd:Q5.
      ?item wdt:P106 wd:Q482980.
      OPTIONAL {
        ?wikipedia_article schema:about ?item.
        ?wikipedia_article schema:isPartOf/wikibase:wikiGroup "wikipedia".
      }
      ?wikiquote_article schema:about ?item.
      ?wikiquote_article schema:isPartOf/wikibase:wikiGroup "wikiquote".
    }
    GROUP BY ?item
    HAVING (?wikiquote_articles > ?wikipedia_articles)
  } AS %sub
  WHERE
  {
    INCLUDE %sub
    SERVICE wikibase:label { bd:serviceParam wikibase:language "en" } .
  }
  ORDER BY DESC(?wikiquote_articles)